Eigenstructure of the Catmull-Clark scheme

Denis Zorin, February 1998

In this worksheet, we examine the eigenstructure of the subdivision matrices of the Catmull-Clark subdivision scheme. This scheme was analyzed in the papers by Ball and Storry (tangent plane continuity) and Peters and Reif (C1-continuity). We mostly follow the analysis found in the latter paper. In addition, we determine the range of coefficients for the extraordinary vertex, for which the scheme is C1-continuous.

Utilities

Subdivision matrix

The constants following Peter and Reif , with K replacing n.

```
> assume( c, real); additionally( c <= 1 ); additionally( c >= -1 );
  CCconst := { p1 = 1/64, p2 = 3/32, p3 = 9/16, q1 = 1/16, q2 = 3/8, r = 1/4 };
  CCvar := { omega = exp( 2*Pi*I*m/K), c = cos( 2*Pi*m/K) };
```

$$CCconst := \{ p1 = \frac{1}{64}, p2 = \frac{3}{32}, p3 = \frac{9}{16}, q1 = \frac{1}{16}, q2 = \frac{3}{8}, r = \frac{1}{4} \}$$

$$CCvar := \{ \omega = \mathbf{e}^{\left(2 \frac{I \pi m}{K}\right)}, c\sim = \cos\left(2 \frac{\pi m}{K}\right) \}$$

The DFT-transformed subdivision matrix; the order of rows is different from P. & R. (there is a typo in Peters and Reif in row 4, elements 1 and 3).

```
> CC := matrix(  [ [alpha*d,    beta*d,                gamma*d,0,0,0,0],
                   [   q2*d, 2*q1*c+q2, q1*(1+conjugate(omega)),0,0,0,0],
                   [    r*d, r*(1+omega),                  r,0,0,0,0],
                   [   p2*d, 2*p1*c+p3,  p2*(1+conjugate(omega)), p2, p1,0,p1*conjugate(omega)],
                   [   q1*d, q1*omega + q2,                q2,                q1, q1,0,0],
                   [   p1*d, p2*(1+omega),                 p3,  p1*(1+omega),   p2, p1, p2 ],
                   [   q1*d, q1 + q2*omega, q2               ,q1*omega,0,0,q1]
  ]);
```

$$CC := \begin{bmatrix} \alpha\, d & \beta\, d & \gamma\, d & 0 & 0 & 0 & 0 \\ q2\, d & 2\, q1\, c\sim + q2 & q1\,(1+\overline{\omega}) & 0 & 0 & 0 & 0 \\ r\, d & r\,(1+\omega) & r & 0 & 0 & 0 & 0 \\ p2\, d & 2\, p1\, c\sim + p3 & p2\,(1+\overline{\omega}) & p2 & p1 & 0 & p1\, \overline{\omega} \\ q1\, d & q1\, \omega + q2 & q2 & q1 & q1 & 0 & 0 \\ p1\, d & p2\,(1+\omega) & p3 & p1\,(1+\omega) & p2 & p1 & p2 \\ q1\, d & q1 + q2\, \omega & q2 & q1\, \omega & 0 & 0 & q1 \end{bmatrix}$$

```
> CCExpanded := map( evalc, subs( { omega = c + s*I, op(CCconst)}, eval(CC))):
```

We are primarily interested in m = 1; in this case d = 0 and we ignore the first row and column

```
> A00 := submatrix( CCExpanded, 2..3,2..3);
```

$$A00 := \begin{bmatrix} \frac{1}{8} c\sim + \frac{3}{8} & \frac{1}{16} + \frac{1}{16} c\sim - \frac{1}{16} I\, s \\ \frac{1}{4} + \frac{1}{4} c\sim + \frac{1}{4} I\, s & \frac{1}{4} \end{bmatrix}$$

```
> A10 := submatrix( CCExpanded, 4..7,2..3):
> A11 := submatrix( CCExpanded, 4..7, 4..7):
> CCZero := map( evalc, subs( { gamma = 1 - alpha - beta, c = 1, omega = 1, d = 1, op(CCconst)}, eval(CC)));
```

$$CCZero := \begin{bmatrix} \alpha & \beta & 1-\alpha-\beta & 0 & 0 & 0 & 0 \\ \frac{3}{8} & \frac{1}{2} & \frac{1}{8} & 0 & 0 & 0 & 0 \\ \frac{1}{4} & \frac{1}{2} & \frac{1}{4} & 0 & 0 & 0 & 0 \\ \frac{3}{32} & \frac{19}{32} & \frac{3}{16} & \frac{3}{32} & \frac{1}{64} & 0 & \frac{1}{64} \\ \frac{1}{16} & \frac{7}{16} & \frac{3}{8} & \frac{1}{16} & \frac{1}{16} & 0 & 0 \\ \frac{1}{64} & \frac{3}{16} & \frac{9}{16} & \frac{1}{32} & \frac{3}{32} & \frac{1}{64} & \frac{3}{32} \\ \frac{1}{16} & \frac{7}{16} & \frac{3}{8} & \frac{1}{16} & 0 & 0 & \frac{1}{16} \end{bmatrix}$$

Eigenvalues

The eigenvalues of the matrix are eigenvalues of $A_{0,0}$ and of $A_{1,1}$.

All eigenvalues of $A_{1,1}$ do not depend on c and are less than 1/8:

```
> map( simplify, subs( s^2 = 1-c^2, [eigenvals( A11)]));
```

$$\left[\frac{1}{64}, \frac{1}{16}, \frac{1}{8}, \frac{1}{32}\right]$$

It can be seen that whenever the eigenvalues are real, one of them is greater than 1/4.

```
> eigenv := map( evalc,  subs( { s^2 = 1 - c^2} , [eigenvals(A00)]));
```

$$eigenv := \left[\frac{1}{16} c\sim + \frac{5}{16} + \frac{1}{16}\sqrt{c\sim^2 + 10\, c\sim + 9}, \frac{1}{16} c\sim + \frac{5}{16} - \frac{1}{16}\sqrt{c\sim^2 + 10\, c\sim + 9}\right]$$

Make sure that the roots are never equal; then the largest is determined simply by comparing them for any value of c.

```
> solve( (op(1, eigenv) - op(2, eigenv))^2 = 0);
```

$$-9, -1$$

Figure out which one is larger.

```
> if simplify( subs( c = 0, op(1, eigenv) > op(2, eigenv)))  then
```

```
    lambda1 := op(1, eigenv);
  else
    lambda1 := op(2, eigenv);
  fi;
```

$$\lambda 1 := \frac{5}{16} + \frac{1}{16}c\sim + \frac{1}{16}\sqrt{9 + 10\,c\sim + c\sim^2}$$

The larger eigenvalue increases with c on the interval of interest. We conclude that the largest eigenvalue for K > 4 is guaranteed to be in the 1st or 0th block of the subdivision matrix. This is also true for K = 3, because there are only two blocks, excluding 0th, and they are complex-conjugate.

```
> solve( {diff(lambda1, c) > 0, abs(c) < 1});
```

$$\{-1 < c\sim, c\sim < 1\}$$

The largest eigenvalues of the 0th block:

```
> evZero := eigenvals( submatrix( CCZero, 1..3,1..3) );
```

$$evZero := 1, \frac{1}{2}\alpha - \frac{1}{8} + \frac{1}{8}\sqrt{16\,\alpha^2 - 8\,\alpha - 3 + 8\,\beta}, \frac{1}{2}\alpha - \frac{1}{8} - \frac{1}{8}\sqrt{16\,\alpha^2 - 8\,\alpha - 3 + 8\,\beta}$$

Determine the ranges for α and β

Determine when the eigenvalues are real.

```
> solve(  (4*( op(2, [evZero]) - op(3, [evZero] )))^2 >= 0 );
```

$$\{\frac{3}{8} - 2\,\alpha^2 + \alpha \le \beta\}$$

Determine where the eigenvalues interesect the level $\lambda_1\left(-\frac{1}{2}\right)$, which is the minimal value of the eigenvalue of the first block, achieved for K = 3.

```
> solve( abs(op(2, [evZero])) = subs( c = -1/2, lambda1)  );
```

$$\{\alpha = \alpha, \beta = -\frac{9}{4}\alpha + \frac{117}{64} - \frac{1}{4}\alpha\sqrt{17} + \frac{13}{64}\sqrt{17}\}, \{\beta = \frac{9}{4}\alpha + \frac{45}{64} + \frac{1}{4}\alpha\sqrt{17} + \frac{5}{64}\sqrt{17}, \alpha = \alpha\}$$

$$\{\beta = -\frac{9}{4}\alpha + \frac{117}{64} - \frac{1}{4}\alpha\sqrt{17} + \frac{13}{64}\sqrt{17}, \alpha = \alpha\}, \{\beta = \frac{9}{4}\alpha + \frac{45}{64} + \frac{1}{4}\alpha\sqrt{17} + \frac{5}{64}\sqrt{17}, \alpha = \alpha\}$$

```
> solve( abs( op(3, [evZero])) = subs( c = -1/2, lambda1)  );
```

$$\{\beta = -\frac{9}{4}\alpha + \frac{117}{64} - \frac{1}{4}\alpha\sqrt{17} + \frac{13}{64}\sqrt{17}, \alpha = \alpha\}, \{\beta = \frac{9}{4}\alpha + \frac{45}{64} + \frac{1}{4}\alpha\sqrt{17} + \frac{5}{64}\sqrt{17}, \alpha = \alpha\}$$

The parabola is the boundary of the region where the eigenvalues are complex.

```
> solve( ( op(2, [evZero]))^2 - (op(3, [evZero]) )^2 = 0 );
```

$$\{\beta = \frac{3}{8} - 2\,\alpha^2 + \alpha, \alpha = \alpha\}, \{\alpha = \frac{1}{4}, \beta = \beta\}$$

The following plot shows the region in the $\alpha - \beta$ plane where the eigenvalues of 0th block are less than $\lambda_1\left(-\frac{1}{2}\right)$. For the coefficients α, β, γ to be positive, α and β have to be inside the triangle [0,0], [1,0], [0,1].

For the magnitudes of eigenvalues to be less than $\lambda_1\left(-\frac{1}{2}\right)$, they have to be in the grey region, the equation of the line delimiting the region is

$$\beta = -\frac{9\,\alpha}{4} + \frac{117}{64} - \frac{1\,\alpha\sqrt{17}}{4} + \frac{13\sqrt{17}}{64}$$

```
> with(plots): display( PLOT(POLYGONS([[1,0], [1,1], [0,1]], COLOR(RGB, 1.0,1.0,1.0), STYLE(PATCH))), contourplot( max(abs(op(3,
  [evZero])), abs(op(2, [evZero]))), alpha = 0..1, beta = 0..1, grid = [20,20], contours = [subs(c = -1/2, lambda1)], filled=true,
  coloring=[grey, white]),  plot( 3/8 - 2*alpha^2 + alpha, alpha = 0..1 , 0..1, color = white), plot( 1-alpha, alpha = 0..1, color =
  black) );
```



Eigenvalue summary. Whenever the coefficients α and β are in the region depicted above, the largest eigenvalue of the subdivision matrix is the eigenvalue of the first block, and it is greater than any other eigenvalue of the subdivison matrix.

Eigenvector

The eigenvector of $A_{0,0}$ can be immediately seen from the matrix.

```
> v0 := [ 4*lambda - 1, (1+ c + I*s) ];
```

$$v0 := [4\lambda - 1, 1 + c\sim + I\, s]$$

Invert $A_{1,.1} - \lambda I$

```
> A11inv := map( simplify, subs( s^2 = 1 - c^2, eval(map( expand, evalm(inverse( evalm(A11 - lambda*&*())))))))):
```

compute v_1 as $-(A_{1,1} - \lambda I)\, A_{1,0}\, v_0$

```
> v1 :=  map( simplify, subs( { s^3 = s*(1-c^2), s^2 = 1-c^2}, map( expand,  evalm( -A11inv &* A10 &* v0 )))):
```

We do not substitute the value for det $A_{1,1} - \lambda I$

```
> CCEigenvector := vector( [ v0[1], v0[2], v1[1], v1[2], v1[3], v1[4]] );
```

$$CCEigenvector := \Bigg[4\lambda - 1, 1 + c\sim + I\, s, 2\, \frac{-60\lambda + 3 + 16\, c\sim \lambda^2 + 20\, c\sim \lambda + 288\lambda^2}{256\lambda^2 - 40\lambda + 1},$$
$$\frac{1024\, c\sim \lambda^3 + 1024\, I\, s\, \lambda^3 + 6144\lambda^3 + 1152\, c\sim \lambda^2 + 1120\, I\, s\, \lambda^2 - 384\lambda^2 - 156\, c\sim \lambda - 96\lambda - 196\, I\, s\, \lambda + 6 + 5\, c\sim + 5\, I\, s}{(256\lambda^2 - 40\lambda + 1)(-1 + 16\lambda)}, 2\,(49152\lambda^4 + 49152\, c\sim \lambda^4 + 49152\, I\, s\, \lambda^4 + 77056\, c\sim \lambda^3 + 256\, c\sim^2 \lambda^3$$
$$+ 76800\, I\, s\, \lambda^3 + 76800\lambda^3 + 256\, I\, s\, \lambda^3 c\sim + 400\, c\sim^2 \lambda^2 + 400\, I\, s\, \lambda^2 c\sim - 11808\lambda^2 - 11808\, I\, s\, \lambda^2 - 11408\, c\sim \lambda^2 + 100\, c\sim^2 \lambda + 60\lambda + 100\, I\, s\, \lambda\, c\sim + 160\, c\sim \lambda + 60\, I\, s\, \lambda + 15 + 15\, c\sim + 15\, I\, s)$$
$$\Big/ ((16384\lambda^3 - 2816\lambda^2 + 104\lambda - 1)(-1 + 16\lambda)),$$
$$\frac{1024\lambda^3 + 6144\, I\, s\, \lambda^3 + 6144\, c\sim \lambda^3 - 384\, I\, s\, \lambda^2 + 32\, c\sim^2 \lambda^2 - 384\, c\sim \lambda^2 + 1120\lambda^2 + 32\, I\, s\, \lambda^2 c\sim - 96\, c\sim \lambda - 96\, I\, s\, \lambda + 40\, I\, s\, \lambda\, c\sim + 40\, c\sim^2 \lambda - 196\lambda + 5 + 6\, I\, s + 6\, c\sim}{-1 + 56\lambda - 896\lambda^2 + 4096\lambda^3} \Bigg]$$

Verify that the vector is correct in the regular case.

```
> map( simplify, subs( { lambda = 1/2,  s = 1, c = 0}, eval(CCEigenvector)));
```

$$[1, 1 + I, 2, 2 + I, 2 + 2\, I, 1 + 2\, I]$$

Code generation